Stewart's Theorem and Apollonius' Theorem

Lukas Bulwahn

April 20, 2020

Abstract

This entry formalizes the two geometric theorems, Stewart's and Apollonius' theorem. Stewart's Theorem [3] relates the length of a triangle's cevian to the lengths of the triangle's two sides. Apollonius' Theorem [2] is a specialisation of Stewart's theorem, restricting the cevian to be the median. The proof applies the law of cosines, some basic geometric facts about triangles and then simply transforms the terms algebraically to yield the conjectured relation. The formalization in Isabelle can closely follow the informal proofs described in the Wikipedia articles of those two theorems.

Contents

1 Stewart's Theorem and Apollonius' Theorem

theory *Stewart-Apollonius*
imports
  Triangle.Triangle
begin

1.1 Stewart's Theorem

theorem *Stewart*:
  fixes $A\ B\ C\ D$ :: $'a$::*euclidean-space*
  assumes *between* $(B,\ C)\ D$
  assumes $a = dist\ B\ C$
  assumes $b = dist\ A\ C$
  assumes $c = dist\ B\ A$
  assumes $d = dist\ A\ D$
  assumes $m = dist\ B\ D$
  assumes $n = dist\ C\ D$
  shows $b^2 * m + c^2 * n = a * (d^2 + m * n)$

proof (*cases*)
 assume $B \neq D \wedge C \neq D$
 let $?\vartheta = angle\ B\ D\ A$
 let $?\vartheta' = angle\ A\ D\ C$
 from ‹$B \neq D \wedge C \neq D$› ‹*between - -*› **have** *cos*: $cos\ ?\vartheta' = -\ cos\ ?\vartheta$
  by (*auto simp add*: *angle-inverse*[*of B C D*] *angle-commute*[*of A D C*])
 from ‹*between - -*› **have** $m + n = a$
  unfolding ‹$a = -$› ‹$m = -$› ‹$n = -$›
  by (*metis* (*no-types*) *between dist-commute*)
 have $c^2 = m^2 + d^2 - 2 * d * m * cos\ ?\vartheta$
  unfolding ‹$c = -$› ‹$m = -$› ‹$d = -$›
  by (*simp add*: *cosine-law-triangle*[*of B A D*] *dist-commute*[*of D A*] *dist-commute*[*of D B*])
 moreover have $b^2 = n^2 + d^2 + 2 * d * n * cos\ ?\vartheta$
  unfolding ‹$b = -$› ‹$n = -$› ‹$d = -$›
  by (*simp add*: *cosine-law-triangle*[*of A C D*] *cos dist-commute*[*of D A*] *dist-commute*[*of D C*])
 ultimately have $b^2 * m + c^2 * n = n * m^2 + n^2 * m + (m + n) * d^2$ **by** *algebra*
 also have $\ldots = (m + n) * (m * n + d^2)$ **by** *algebra*
 also from ‹$m + n = a$› **have** $\ldots = a * (d^2 + m * n)$ **by** *simp*
 finally show *?thesis* **.**
next
 assume $\neg (B \neq D \wedge C \neq D)$
 from *this assms* **show** *?thesis* **by** (*auto simp add*: *dist-commute*)
qed

Here is an equivalent formulation that is probably more suitable for further use in other geometry theories in Isabelle.

theorem *Stewart'*:
 fixes $A\ B\ C\ D$ :: *'a::euclidean-space*
 assumes *between* $(B, C)\ D$
 shows $(dist\ A\ C)^2 * dist\ B\ D + (dist\ B\ A)^2 * dist\ C\ D = dist\ B\ C * ((dist\ A\ D)^2 + dist\ B\ D * dist\ C\ D)$
using *assms* **by** (*auto intro*: *Stewart*)

1.2 Apollonius' Theorem

Apollonius' theorem is a simple specialisation of Stewart's theorem, but historically predated Stewart's theorem by many centuries.

lemma *Apollonius*:
 fixes $A\ B\ C$ :: *'a::euclidean-space*
 assumes $B \neq C$
 assumes $b = dist\ A\ C$
 assumes $c = dist\ B\ A$
 assumes $d = dist\ A\ (midpoint\ B\ C)$
 assumes $m = dist\ B\ (midpoint\ B\ C)$
 shows $b^2 + c^2 = 2 * (m^2 + d^2)$

proof −
  from ‹$B \neq C$› **have** $m \neq 0$
    unfolding ‹m = -› **using** *midpoint-eq-endpoint*(*1*) **by** *fastforce*
  have *between* (B, C) (*midpoint B C*)
    by (*simp add*: *between-midpoint*)
  moreover have *dist C* (*midpoint B C*) = *dist B* (*midpoint B C*)
    by (*simp add*: *dist-midpoint*)
  moreover have *dist B C* = *2* ∗ *dist B* (*midpoint B C*)
    by (*simp add*: *dist-midpoint*)
  moreover note *assms*(*2*−*5*)
  ultimately have $b^2 * m + c^2 * m = (2 * m) * (m^2 + d^2)$
    by (*auto dest*!: *Stewart*[**where** *a*=*2* ∗ *m*] *simp add*: *power2-eq-square*)
  from *this* **have** $m * (b^2 + c^2) = m * (2 * (m^2 + d^2))$
    by (*simp add*: *distrib-left semiring-normalization-rules*(*7*))
  from *this* ‹$m \neq 0$› **show** *?thesis* **by** *auto*
qed

Here is the equivalent formulation that is probably more suitable for further use in other geometry theories in Isabelle.

lemma *Apollonius'*:
  fixes *A B C* :: *'a*::*euclidean-space*
  assumes $B \neq C$
  shows $(dist\ A\ C)^2 + (dist\ B\ A)^2 = 2 * ((dist\ B\ (midpoint\ B\ C))^2 + (dist\ A\ (midpoint\ B\ C))^2)$
using *assms* **by** (*rule Apollonius*) *auto*

end

References

[1] D. B. Surowski. Advanced high-school mathematics, 2011. https://www.math.ksu.edu/~dbski/writings/further.pdf [Online; accessed 30-July-2017].

[2] Wikipedia. Apollonius' theorem — wikipedia, the free encyclopedia, 2017. https://en.wikipedia.org/w/index.php?title=Apollonius%27_theorem&oldid=790659235 [Online; accessed 30-July-2017].

[3] Wikipedia. Stewart's theorem — wikipedia, the free encyclopedia, 2017. https://en.wikipedia.org/w/index.php?title=Stewart%27s_theorem&oldid=790777285 [Online; accessed 30-July-2017].